
4.1 Change 1
	Testcase name
	6.2.3.1

	Reason for change
	Testcase 6.2.3.1 uses 2 GERAN Cells. The test case fails when the SS, after receiving a RAU request on Cell 2, sends the Authentication & Ciphering Request.This  message is sent using the G_LLC_UNITDATA_REQ primitive which uses the LLMEID (Logical Layer Management Entity Id) instead of the CellId.The mapping between the LLMEID and the CellId is defined through the G_CLLC_CreateLLE_REQ primitive.In the ETSI test cases the G_CLLC_CreateLLE_REQ primitives for Cell1 and Cell2 are sent when the two cells are created and they map the same LLMEID =1 both to Cell1 and Cell2.

This way, when the G_LLC_UNITDATA_REQ primitive is sent (using LLMEID=1), it is not clear how can the SS determine if it is related to Cell1 or Cell2.

The same LLMEID should not be mapped at the same time to two different cells.

	Summary of change
	Add parameter p_LLMEId to templates

	Source of change
	\CommonGERAN\ GERAN_Templates.ttcn


Before change:

	…
  template G_LLC_DATAMESSAGE_IND car_G_LLC_UnitData_IND(template LLCUnitDataIndType p_Msg,

                                                        template (present) integer p_SAPI := ?) :=

  { /* The ASP is used to receive an L3 PDU from the UE/MS in LLC unconfirmed transmission.

       3GPP TS 44.064 clause 8.4.2 */

    /* @status    APPROVED */

    unit := {

      lLMEId          := 1,
      tLLI            := ?,                            // temporary logical link identifier

      sAPI            := p_SAPI,

      msg             := p_Msg                         // L3 PDU

    }

  };

//---------------------------------------------------------------------------------------------------------------------
…

  template (value) G_LLC_DATAMESSAGE_REQ cas_G_LLC_UnitData_Req(TLLI p_TLLI,

                                                                SAPI p_Sapi,

                                                                B1_Type p_ProtectMode,

                                                                B1_Type p_CipherMode,

                                                                template (value) LLCUnitDataReqType p_Msg) :=

  { /* The ASP is used to transmit an L3 PDU to the UE/MS in LLC unconfirmed transmission.

       3GPP TS 44.064 clause 8.4.1 */

    /* @status    APPROVED */

    unit := {

      lLMEId          := 1,

      tLLI            := p_TLLI,                       // temporary logical link identifier

      sAPI            := p_Sapi,

      protectMode     := p_ProtectMode,                // 0 -- unprotected; 1 -- protected

      cipherMode      := p_CipherMode,                 // 0 -- no encryption; 1 -- encrypted

      msg             := p_Msg                         // L3 PDU

    }

  };
…

//---------------------------------------------------------------------------------------------------------------------
…

  template (value) G_LLC_CONFIG_REQ cas_G_CLLC_CreateLLE_REQ(GERAN_CellId_Type p_CellId) :=

  { /* The ASP is used to create an LLE (LLC Entity) in GERAN LLC emulation module.

       The RlcMacEntityId needs to be created prior to this by G_CRLC_CreateRLC_MAC_REQ ASP. */

    /* @status    APPROVED */

    createLLC := {

      lLMEId          := 1,                       // Logical Layer Management Entity Id

      rLC_MAC_MappingInfo := enum2int(p_CellId)   // This parameter indicates the RLC/MAC emulation module in the cell, not the cell itself.

    }

  };

  //-----------------------------------------------------------------------------

  template (value) G_LLC_CONFIG_REQ cas_G_CLLC_DeleteLLE_REQ :=

  { /* The ASP is used to delete an LLE (LLC Entity) in GERAN LLC emulation module.

       The RlcMacEntityId needs to be created prior to this by G_CRLC_CreateRLC_MAC_REQ ASP. */

    deleteLLC := {

      lLMEId  := 1                      // Logical Layer Management Entity Id

    }

  };




After change:

	…
  template G_LLC_DATAMESSAGE_IND car_G_LLC_UnitData_IND(template LLCUnitDataIndType p_Msg,

                                                        template (present) integer p_SAPI := ?) :=

  { /* The ASP is used to receive an L3 PDU from the UE/MS in LLC unconfirmed transmission.

       3GPP TS 44.064 clause 8.4.2 */

    /* @status    APPROVED */

    unit := {

      lLMEId          := ?, 

      tLLI            := ?,                            // temporary logical link identifier

      sAPI            := p_SAPI,

      msg             := p_Msg                         // L3 PDU

    }

  };

//---------------------------------------------------------------------------------------------------

  template (value) G_LLC_DATAMESSAGE_REQ cas_G_LLC_UnitData_Req(TLLI p_TLLI,

                                                                SAPI p_Sapi,

                                                                B1_Type p_ProtectMode,

                                                                B1_Type p_CipherMode,

                                                                template (value) LLCUnitDataReqType p_Msg,

                                                                integer p_LLMEId :=1) :=       
  { /* The ASP is used to transmit an L3 PDU to the UE/MS in LLC unconfirmed transmission.

       3GPP TS 44.064 clause 8.4.1 */

    /* @status    APPROVED */

    unit := {

      lLMEId          := p_LLMEId,   

      tLLI            := p_TLLI,                       // temporary logical link identifier

      sAPI            := p_Sapi,

      protectMode     := p_ProtectMode,                // 0 -- unprotected; 1 -- protected

      cipherMode      := p_CipherMode,                 // 0 -- no encryption; 1 -- encrypted

      msg             := p_Msg                         // L3 PDU

    }

  };

…

//---------------------------------------------------------------------------------------------------------------------
…

  template (value) G_LLC_CONFIG_REQ cas_G_CLLC_CreateLLE_REQ(GERAN_CellId_Type p_CellId, integer p_LLMEId :=1 ) := 
  { /* The ASP is used to create an LLE (LLC Entity) in GERAN LLC emulation module.

       The RlcMacEntityId needs to be created prior to this by G_CRLC_CreateRLC_MAC_REQ ASP. */

    /* @status    APPROVED */

    createLLC := {

      lLMEId  := p_LLMEId,                         // Logical Layer Management Entity Id

      rLC_MAC_MappingInfo := enum2int(p_CellId)   // This parameter indicates the RLC/MAC emulation module in the cell, not the cell itself.

    }

  };

  //-----------------------------------------------------------------------------

  template (value) G_LLC_CONFIG_REQ cas_G_CLLC_DeleteLLE_REQ(integer p_LLMEId:=1):=  //Anritsu TTCN CR

  { /* The ASP is used to delete an LLE (LLC Entity) in GERAN LLC emulation module.

       The RlcMacEntityId needs to be created prior to this by G_CRLC_CreateRLC_MAC_REQ ASP. */

    deleteLLC := {

      lLMEId  := p_LLMEId                      // Logical Layer Management Entity Id

    }

  };


	


4.2 Change 2

	Testcase name
	6.2.3.1

	Reason for change
	Need to assign the LLMEIds based on GERAN cell ID


	Summary of change
	Add different values for the parameter of p_LLMEId 

	Source of change
	\CommonGERAN\GERAN_CommonFunctions.ttcn


Before change:

	…

  function f_Create_LLE_Entity(GERAN_CellId_Type p_CellId) runs on GERAN_PTC

  { /* @sic R5s120050 MCC160 implementation sic@ */

    f_GERAN_LLC_CONFIG_Common(cas_G_CLLC_CreateLLE_REQ (p_CellId));

  }
//----------------------------------------------------------------------------
…

  function f_Delete_GPRS_Entities (GERAN_CellId_Type p_CellId) runs on GERAN_PTC

  {

    f_GERAN_RLC_CONFIG_Common(cas_G_CRLC_DeleteRLC_MAC_REQ (p_CellId));

    f_GERAN_LLC_CONFIG_Common(cas_G_CLLC_DeleteLLE_REQ);

  } //end of f_Delete_GPRS_Entities

 //----------------------------------------------------------------------------
…

   function f_GERAN_GPRSAttach(GERAN_CellId_Type p_CellId, boolean p_DoAuth := true, boolean p_CheckEUTRA := false) runs on GERAN_PTC

  {

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var ATTACHREQUEST v_AttachRequest;

    var B3_Type v_AttachType;

    var octetstring v_PLMN := f_GERAN_PLMN_Get(p_CellId);

    var O1_Type v_RAC := f_GERAN_RAC_Get(p_CellId);

    var O2_Type v_LAC := f_GERAN_LAC_Get(p_CellId);

    var B1_Type v_Ciph :=  '0'B;

    // Receive GPRS Attach Request Message

    if (p_CheckEUTRA) { // GP-110050

      v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell24, car_G_LLC_UnitData_IND (cr_G_AttachReq_EUTRANCap (cr_AttachTypeAny,

                                                                                                               cr_MobileIdAny (omit),

                                                                                                               cr_RAI (omit),

                                                                                                               ?)));

      v_AttachRequest := v_LLCDataInd.unit.msg.attReq;

      if (not ((v_AttachRequest.msRadioAccessCap.valuePart.listofValues[0].raCapability.singleAccess.eutraFDD == '1'B) or

               (v_AttachRequest.msRadioAccessCap.valuePart.listofValues[0].raCapability.singleAccess.eutraTDD == '1'B)) and

          not ((v_AttachRequest.msRadioAccessCap.valuePart.listofValues[0].raCapability.singleAccess.geran2EutraSupport == '10'B) or

               (v_AttachRequest.msRadioAccessCap.valuePart.listofValues[0].raCapability.singleAccess.geran2EutraSupport == '11'B)) and

          not ((v_AttachRequest.msClassmark3.eutraFDDSupport == '1'B) or

               (v_AttachRequest.msClassmark3.eutraFDDSupport == '1'B))) {

        f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE doesn't support EUTRA");

      }

    } else {

      v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (p_CellId, car_G_LLC_UnitData_IND (cr_G_AttachReq (cr_AttachTypeAny,

                                                                                                 cr_MobileIdAny (omit),

                                                                                                 cr_RAI (omit),

                                                                                                 ?)));

    }

    v_AttachType := v_LLCDataInd.unit.msg.attReq.attachType.attachType;

    f_GERAN_SplitPGCycle_Set(p_CellId, v_LLCDataInd.unit.msg.attReq.drxParameter);

    if (p_DoAuth) {

      f_GERAN_GPRS_Authentication (p_CellId);

      v_Ciph := px_GSM_CipheringOnOff;

    }

    // Send GPRS Attach Accept

    f_GPRS_SendDownlinkMsg(p_CellId, 1, cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(p_CellId),

                                                                tsc_LLCSapi_GMM,

                                                                tsc_LLC_PM,

                                                                v_Ciph, // @sic R5s110593 sic@

                                                                cs_G_AttachAccept (cs_GMM_AttachResult(v_AttachType),

                                                                                   cs_RAI(omit,

                                                                                          cs_LAI(omit, v_PLMN, v_LAC),

                                                                                          v_RAC),

                                                                                   cs_PTMSI_Signature (px_PTMSI_SigDef),

                                                                                   cs_MobileIdTMSIorPTMSI ('18'O, px_PTMSI_Def))));

    //Receive GPRS Attach Complete Message

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (p_CellId, car_G_LLC_UnitData_IND (cr_G_AttachComplete));

  }// end of GPRSAttach

//----------------------------------------------------------------------------
…

  function f_GERAN_GPRS_Authentication (GERAN_CellId_Type p_CellId) runs on GERAN_PTC

  {

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var TLLI v_TLLI;

    var GERAN_SecurityInfo_Type v_SecurityInfo := f_GERAN_SecurityInfo_Get();

    var B32_Type    v_AuthRES_High;

    var bitstring   v_AuthRES;

    v_SecurityInfo.authPS := f_AuthenticationInit (v_SecurityInfo.authPS);

    f_GERAN_SecurityInfo_Set(v_SecurityInfo);  // @sic R5s110596 sic@

    // Send AUTHENTICATION AND CIPHERING REQUEST using relevant PS keys computed before.

    f_GPRS_SendDownlinkMsg(p_CellId, 1, cas_G_LLC_UnitData_Req ( f_GERAN_TLLI_Get(p_CellId),

                                                                 tsc_LLCSapi_GMM,

                                                                 tsc_LLC_PM,

                                                                 tsc_LLC_NoCiph,

                                                                 cs_G_AuthAndCiphReq (cs_GMM_AuthRAND(v_SecurityInfo.authPS.RandValue),

                                                                                      cs_GMM_KeySeq(v_SecurityInfo.authPS.KeySeq),

                                                                                      v_SecurityInfo.gprsCipherAlg,

                                                                                      cs_GMM_AUTN(v_SecurityInfo.authPS.AUTN))));

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (p_CellId, car_G_LLC_UnitData_IND

                                             (( cr_G_AuthAndCiphRsp (cr_AuthRspAny, cr_AuthCiphRspExtAny),

                                                cr_G_AuthAndCiphRsp (cr_AuthRspAny, omit),

                                                cr_G_AuthAndCiphFailureAny)));

    if (match (v_LLCDataInd, car_G_LLC_UnitData_IND ( cr_G_AuthAndCiphRsp (cr_AuthRspAny, cr_AuthCiphRspExtAny))))

      {// AUTHENTICATION AND CIPHERING RESPONSE including both Authentication Response parameters (RES and RES ext)

        v_AuthRES_High := v_LLCDataInd.unit.msg.authCipherResp.authRsp.rES;

        v_AuthRES := v_AuthRES_High & v_LLCDataInd.unit.msg.authCipherResp.authRspExt.rES;

        // don't change the TLLI, only the ciphering algorithm

        v_TLLI := f_GERAN_TLLI_Get(p_CellId); // careful, because this might have changed in ULTBF

        f_LLC_TLLI_Assign(p_CellId, v_TLLI, v_TLLI);

        // Verify that the received Authentication Response parameters match expected response:

        if (v_AuthRES != substr(v_SecurityInfo.authPS.XRES, 0, lengthof(v_AuthRES))) {

          f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Authentication Failed");

        }

      }

    else if (match (v_LLCDataInd, car_G_LLC_UnitData_IND (cr_G_AuthAndCiphRsp (cr_AuthRspAny, omit))))

      {// AUTHENTICATION AND CIPHERING RESPONSE including only one Authentication Response parameter (RES)

        v_AuthRES_High := v_LLCDataInd.unit.msg.authCipherResp.authRsp.rES;

        // don't change the TLLI, only the ciphering algorithm

        v_TLLI := f_GERAN_TLLI_Get(p_CellId); // careful, because this might have changed in ULTBF

        f_LLC_TLLI_Assign(p_CellId, v_TLLI, v_TLLI);

        // Verify that the received Authentication Response parameters match expected response.

        if (v_AuthRES_High != substr(v_SecurityInfo.authPS.XRES, 0, lengthof(v_AuthRES_High))) {

          f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Authentication Failed");

        }

      }

    else if (match (v_LLCDataInd, car_G_LLC_UnitData_IND ( cr_G_AuthAndCiphFailureAny)))

      { //AUTHENTICATION AND CIPHERING Failure, so stop execution

        f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Received Authentication and Ciphering Failure");

      }

    else { // shouldn't get here because it should match with one of the above 3

      f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Unexpected message");

    }

  }//end of f_GPRS_Authentication

  


After change:

	…

  function f_Create_LLE_Entity(GERAN_CellId_Type p_CellId) runs on GERAN_PTC

  { /* @sic R5s120050 MCC160 implementation sic@ */

    var integer v_LLMEId := enum2int(p_CellId);  

    f_GERAN_LLC_CONFIG_Common(cas_G_CLLC_CreateLLE_REQ (p_CellId, v_LLMEId)); 

  }

  //----------------------------------------------------------------------------
…

  function f_Delete_GPRS_Entities (GERAN_CellId_Type p_CellId) runs on GERAN_PTC

  {

    var integer v_LLMEId := enum2int(p_CellId); 

    f_GERAN_RLC_CONFIG_Common(cas_G_CRLC_DeleteRLC_MAC_REQ (p_CellId));

    f_GERAN_LLC_CONFIG_Common(cas_G_CLLC_DeleteLLE_REQ (v_LLMEId));  

  } //end of f_Delete_GPRS_Entities

//----------------------------------------------------------------------------
…

  function f_GERAN_GPRSAttach(GERAN_CellId_Type p_CellId, boolean p_DoAuth := true, boolean p_CheckEUTRA := false) runs on GERAN_PTC

  {

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var ATTACHREQUEST v_AttachRequest;

    var B3_Type v_AttachType;

    var octetstring v_PLMN := f_GERAN_PLMN_Get(p_CellId);

    var O1_Type v_RAC := f_GERAN_RAC_Get(p_CellId);

    var O2_Type v_LAC := f_GERAN_LAC_Get(p_CellId);

    var B1_Type v_Ciph :=  '0'B;

    var integer v_LLMEId := enum2int(p_CellId); 
    // Receive GPRS Attach Request Message

    if (p_CheckEUTRA) { // GP-110050

      v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell24, car_G_LLC_UnitData_IND (cr_G_AttachReq_EUTRANCap (cr_AttachTypeAny,

                                                                                                               cr_MobileIdAny (omit),

                                                                                                               cr_RAI (omit),

                                                                                                               ?)));

      v_AttachRequest := v_LLCDataInd.unit.msg.attReq;

      if (not ((v_AttachRequest.msRadioAccessCap.valuePart.listofValues[0].raCapability.singleAccess.eutraFDD == '1'B) or

               (v_AttachRequest.msRadioAccessCap.valuePart.listofValues[0].raCapability.singleAccess.eutraTDD == '1'B)) and

          not ((v_AttachRequest.msRadioAccessCap.valuePart.listofValues[0].raCapability.singleAccess.geran2EutraSupport == '10'B) or

               (v_AttachRequest.msRadioAccessCap.valuePart.listofValues[0].raCapability.singleAccess.geran2EutraSupport == '11'B)) and

          not ((v_AttachRequest.msClassmark3.eutraFDDSupport == '1'B) or

               (v_AttachRequest.msClassmark3.eutraFDDSupport == '1'B))) {

        f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE doesn't support EUTRA");

      }

    } else {

      v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (p_CellId, car_G_LLC_UnitData_IND (cr_G_AttachReq (cr_AttachTypeAny,

                                                                                                 cr_MobileIdAny (omit),

                                                                                                 cr_RAI (omit),

                                                                                                 ?)));

    }

    v_AttachType := v_LLCDataInd.unit.msg.attReq.attachType.attachType;

    f_GERAN_SplitPGCycle_Set(p_CellId, v_LLCDataInd.unit.msg.attReq.drxParameter);

    if (p_DoAuth) {

      f_GERAN_GPRS_Authentication (p_CellId);

      v_Ciph := px_GSM_CipheringOnOff;

    }

    // Send GPRS Attach Accept

    f_GPRS_SendDownlinkMsg(p_CellId, 1, cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(p_CellId),

                                                                tsc_LLCSapi_GMM,

                                                                tsc_LLC_PM,

                                                                v_Ciph, // @sic R5s110593 sic@

                                                                cs_G_AttachAccept (cs_GMM_AttachResult(v_AttachType),

                                                                                   cs_RAI(omit,

                                                                                          cs_LAI(omit, v_PLMN, v_LAC),

                                                                                          v_RAC),

                                                                                   cs_PTMSI_Signature (px_PTMSI_SigDef),

                                                                                   cs_MobileIdTMSIorPTMSI ('18'O, px_PTMSI_Def)),v_LLMEId)); 

    //Receive GPRS Attach Complete Message

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (p_CellId, car_G_LLC_UnitData_IND (cr_G_AttachComplete));

  }// end of GPRSAttach

//----------------------------------------------------------------------------
…

  function f_GERAN_GPRS_Authentication (GERAN_CellId_Type p_CellId) runs on GERAN_PTC

  {

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var TLLI v_TLLI;

    var GERAN_SecurityInfo_Type v_SecurityInfo := f_GERAN_SecurityInfo_Get();

    var B32_Type    v_AuthRES_High;

    var bitstring   v_AuthRES;

    var integer v_LLMEId := enum2int(p_CellId); 
    v_SecurityInfo.authPS := f_AuthenticationInit (v_SecurityInfo.authPS);

    f_GERAN_SecurityInfo_Set(v_SecurityInfo);  // @sic R5s110596 sic@

    // Send AUTHENTICATION AND CIPHERING REQUEST using relevant PS keys computed before.

    f_GPRS_SendDownlinkMsg(p_CellId, 1, cas_G_LLC_UnitData_Req ( f_GERAN_TLLI_Get(p_CellId),

                                                                 tsc_LLCSapi_GMM,

                                                                 tsc_LLC_PM,

                                                                 tsc_LLC_NoCiph,

                                                                 cs_G_AuthAndCiphReq (cs_GMM_AuthRAND(v_SecurityInfo.authPS.RandValue),

                                                                                      cs_GMM_KeySeq(v_SecurityInfo.authPS.KeySeq),

                                                                                      v_SecurityInfo.gprsCipherAlg,

                                                                                      cs_GMM_AUTN(v_SecurityInfo.authPS.AUTN)),v_LLMEId));
    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (p_CellId, car_G_LLC_UnitData_IND

                                             (( cr_G_AuthAndCiphRsp (cr_AuthRspAny, cr_AuthCiphRspExtAny),

                                                cr_G_AuthAndCiphRsp (cr_AuthRspAny, omit),

                                                cr_G_AuthAndCiphFailureAny)));

    if (match (v_LLCDataInd, car_G_LLC_UnitData_IND ( cr_G_AuthAndCiphRsp (cr_AuthRspAny, cr_AuthCiphRspExtAny))))

      {// AUTHENTICATION AND CIPHERING RESPONSE including both Authentication Response parameters (RES and RES ext)

        v_AuthRES_High := v_LLCDataInd.unit.msg.authCipherResp.authRsp.rES;

        v_AuthRES := v_AuthRES_High & v_LLCDataInd.unit.msg.authCipherResp.authRspExt.rES;

        // don't change the TLLI, only the ciphering algorithm

        v_TLLI := f_GERAN_TLLI_Get(p_CellId); // careful, because this might have changed in ULTBF

        f_LLC_TLLI_Assign(p_CellId, v_TLLI, v_TLLI);

        // Verify that the received Authentication Response parameters match expected response:

        if (v_AuthRES != substr(v_SecurityInfo.authPS.XRES, 0, lengthof(v_AuthRES))) {

          f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Authentication Failed");

        }

      }

    else if (match (v_LLCDataInd, car_G_LLC_UnitData_IND (cr_G_AuthAndCiphRsp (cr_AuthRspAny, omit))))

      {// AUTHENTICATION AND CIPHERING RESPONSE including only one Authentication Response parameter (RES)

        v_AuthRES_High := v_LLCDataInd.unit.msg.authCipherResp.authRsp.rES;

        // don't change the TLLI, only the ciphering algorithm

        v_TLLI := f_GERAN_TLLI_Get(p_CellId); // careful, because this might have changed in ULTBF

        f_LLC_TLLI_Assign(p_CellId, v_TLLI, v_TLLI);

        // Verify that the received Authentication Response parameters match expected response.

        if (v_AuthRES_High != substr(v_SecurityInfo.authPS.XRES, 0, lengthof(v_AuthRES_High))) {

          f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Authentication Failed");

        }

      }

    else if (match (v_LLCDataInd, car_G_LLC_UnitData_IND ( cr_G_AuthAndCiphFailureAny)))

      { //AUTHENTICATION AND CIPHERING Failure, so stop execution

        f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Received Authentication and Ciphering Failure");

      }

    else { // shouldn't get here because it should match with one of the above 3

      f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Unexpected message");

    }

  }//end of f_GPRS_Authentication

  //----------------------------------------------------------------------------

