
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of EUTRA-GERAN test case 6.2.3.24  which is part of the LTE/SAE test suite ‘iwd-EUTRA-B2012-03_D12wk24’. 

The test case can be demonstrated to run with one LTE UE (see section 5). Execution log is provided as evidence.

3. Verification Test Summary

Test Case:
TC_6_2_3_24
Test Group:
6_2/InterRat_CellReSelection_GtoE_GERAN.ttcn

ATS Version:
iwd-EUTRA-B2012-03_D12wk24
System Simulator used:
R&S CMW500 
UE used:
Nvidia  ICERA 410
Verification Status:
PASS


4. Corrections 

Change 1 – Correction to ‘f_TC_6_2_3_24_GERAN ()’
	Function name
	f_TC_6_2_3_24_GERAN ()

	Reason for change
	1. According to TS 38.508 secn 4.4.5, the default value for parameter 3G Measurement Parameters Description should be “Not Present”. This IE should be set to Omit in the SI 2quater for 6.2.3.24
2. According to TS 24.008 section 9.4.1, Mobile Station Classmark 3 is an Optional IE. A conformant UE can choose not to send this IE. However, the prose for TC 6.2.3.24 mandates that the IEs E-UTRA FDD Support and E-UTRA TDD Support are checked which both are part of Mobile Station Classmark 3.
3. At Step 4 of the prose, PACKET CELL CHANGE ORDER message is sent to the UE. As the uplink TBF is already in operation for on-going UL Data Transfer, the PCCO can be sent using this UL TBF

	Summary of change
	1. 3G Measurement Parameters Description is set to Omit.
2. Function f_GERAN_Preamble is called with one parameter. A prose CR shall be raised for this change (see “draft_GP-12nnnn_...” in R5s120522). 
3. Changed the TBF direction from downlink (0) to uplink (1)

	TTCN module
	6_2\InterRat_CellReSelection_GtoE_GERAN.ttcn

	MCC160 Comment
	


Before change

	...

  function f_TC_6_2_3_24_GERAN ( ) runs on GERAN_PTC
  {
    var IRAT_CoOrd_SysInfo_Type v_EUTRA_SysInfo;
    var template (value) RptdEUTRANNeighCells v_RptdEUTRANNeighCells;
    var template(value) SYSTEMINFORMATIONTYPE2quater v_SI2quater;
    var boolean v_EndOfTBF := false;
    //* @desc Initialise all cells
    f_GERAN_Init (EUTRA_GERAN);
    //* @desc Wait for EUTRA cell information to configure in GERAN cell
    v_EUTRA_SysInfo := f_IRAT_WaitForCoOrd_SysInfo (EUTRA);
    //Configure Repeated E-UTRA Neighbour Cells specific message contents,
    //change SI2q according to test case
    v_RptdEUTRANNeighCells := {
      cs_RptdEUTRAN_NeighCell(cs_RptdEARFCNList1(int2bit(v_EUTRA_SysInfo.Eutra[0].Arfcn, 16),
                                                 f_GERAN_ConvertIntegerToEUTRA_MeasurementBandwidth (v_EUTRA_SysInfo.Eutra[0].DL_Bandwidth)),
                              '100'B,
                              '00010'B,
                              '00010'B,
                              '10001'B),
      cs_RptdEUTRAN_NeighCellEmpty
    };
    v_SI2quater := cs_SysInfoType2quater(cds_SI2quaterRO_ServingCellParams('0000'B,
                                                                           '0000'B,
                                                                           '0'B,
                                                                           '0'B,
                                                                           omit,
                                                                           '0'B,
                                                                           omit,
                                                                           '1'B,
                                                                           cs_SI2quaterMeasParams3G_ISHO,
                                                                           '0'B,
                                                                           omit,
                                                                           cs_EUTRAN_MeasParams_SI2(v_RptdEUTRANNeighCells,
                                                                                                               '1'B), // CCN Active
                                                                           '1'B, // @sic R5s120021 sic@
                                                                           '001'B, //geranPrior
                                                                           '1111'B, //ThreshPriorSearch
                                                                           '0000'B, //threshGSMLow
                                                                           '00'B, //HPrior
                                                                           '00'B)); //TResel
    // Set the SI2 with the information provided from E-UTRA
    f_GERAN_SysInfo_SetSI2q_1 (geran_Cell24, v_SI2quater);
    f_GERAN_SysInfo_SetSI2q_2 (geran_Cell24, omit); //R5s120129 Add. Ch. 1
    // Create GERAN cell and broadcast modified System Infomation
    f_GERAN_CreateCell (geran_Cell24);
    // Active PDP context 2 on GERAN cell
    f_GERAN_Preamble (geran_Cell24, true);
    f_GERAN_ActivatePDPContext (geran_Cell24);
    // in case the UE sends an XID once the bearer is established
    f_GERAN_LLC_XID (geran_Cell24, false);
    f_GERAN_TestBody_Set( true );
    //@siclog "Step 1" siclog@
    f_GERAN_Trigger_UplinkData (geran_Cell24, 10000);
    //@siclog "Step 2" siclog@
    // Now tell the EUTRA component to adjust power level
    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);
    //@siclog "Step 3" siclog@
    alt {
      [] a_GERAN_Receive_UplinkData (geran_Cell24, v_EndOfTBF) {}
      [] G_RLC.receive (car_G_RLC_ControlMsg_IND(geran_Cell24, tsc_PhyCh1, cr_PktCellChangeNotificationAny)){}
    }
    //@siclog "Step 4" siclog@
    G_RLC.send (cas_G_RLC_ControlMsg_REQ_PACCH(geran_Cell24,
                                               tsc_PhyCh1,
                                               0,                                         

                                               tsc_UplinkTFI,
                                               cs_PktCellChangeOrder (cs_GlobalTFIul,
                                                                      cs_EUTRAN_TargetCell (int2bit(v_EUTRA_SysInfo.Eutra[0].Arfcn, 16),
                                                                                            '0'B,
                                                                                            omit,
                                                                                            int2bit (v_EUTRA_SysInfo.Eutra[0].HO.PhysCellId, 9)))));
    //@siclog "Step 5" siclog@
    // Tell the EUTRA component to wait for an RRC Conn Req
    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);
    // Keep going until UE is on cell 1
    alt {
      [not v_EndOfTBF] a_GERAN_Receive_UplinkData (geran_Cell24, v_EndOfTBF) {}
      [] EUTRA.receive(cmr_IRAT_Trigger) {} // Trigger received, so can finish loop now
    }
    // Test now finished
    f_GERAN_TestBody_Set ( false );
    f_GERAN_SS_CellRelease ( geran_Cell24 );
  }
...


After change

	...

function f_TC_6_2_3_24_GERAN ( ) runs on GERAN_PTC
  {
    var IRAT_CoOrd_SysInfo_Type v_EUTRA_SysInfo;
    var template (value) RptdEUTRANNeighCells v_RptdEUTRANNeighCells;
    var template(value) SYSTEMINFORMATIONTYPE2quater v_SI2quater;
    var boolean v_EndOfTBF := false;
    //* @desc Initialise all cells
    f_GERAN_Init (EUTRA_GERAN);
    //* @desc Wait for EUTRA cell information to configure in GERAN cell
    v_EUTRA_SysInfo := f_IRAT_WaitForCoOrd_SysInfo (EUTRA);
    //Configure Repeated E-UTRA Neighbour Cells specific message contents,
    //change SI2q according to test case
    v_RptdEUTRANNeighCells := {
      cs_RptdEUTRAN_NeighCell(cs_RptdEARFCNList1(int2bit(v_EUTRA_SysInfo.Eutra[0].Arfcn, 16),
                                                 f_GERAN_ConvertIntegerToEUTRA_MeasurementBandwidth (v_EUTRA_SysInfo.Eutra[0].DL_Bandwidth)),
                              '100'B,
                              '00010'B,
                              '00010'B,
                              '10001'B),
      cs_RptdEUTRAN_NeighCellEmpty
    };
    v_SI2quater := cs_SysInfoType2quater(cds_SI2quaterRO_ServingCellParams('0000'B,
                                                                           '0000'B,
                                                                           '0'B,
                                                                           '0'B,
                                                                           omit,
                                                                           '0'B,
                                                                           omit,
                                                                           '0'B,
                                                                           omit,
                                                                           '0'B,
                                                                           omit,
                                                                           cs_EUTRAN_MeasParams_SI2(v_RptdEUTRANNeighCells,
                                                                                                               '1'B), // CCN Active
                                                                           '1'B, // @sic R5s120021 sic@
                                                                           '001'B, //geranPrior
                                                                           '1111'B, //ThreshPriorSearch
                                                                           '0000'B, //threshGSMLow
                                                                           '00'B, //HPrior
                                                                           '00'B)); //TResel
    // Set the SI2 with the information provided from E-UTRA
    f_GERAN_SysInfo_SetSI2q_1 (geran_Cell24, v_SI2quater);
    f_GERAN_SysInfo_SetSI2q_2 (geran_Cell24, omit); //R5s120129 Add. Ch. 1
    // Create GERAN cell and broadcast modified System Infomation
    f_GERAN_CreateCell (geran_Cell24);
    // Active PDP context 2 on GERAN cell
    f_GERAN_Preamble (geran_Cell24);
    f_GERAN_ActivatePDPContext (geran_Cell24);
    // in case the UE sends an XID once the bearer is established
    f_GERAN_LLC_XID (geran_Cell24, false);
    f_GERAN_TestBody_Set( true );
    //@siclog "Step 1" siclog@
    f_GERAN_Trigger_UplinkData (geran_Cell24, 10000);
    //@siclog "Step 2" siclog@
    // Now tell the EUTRA component to adjust power level
    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);
    //@siclog "Step 3" siclog@
    alt {
      [] a_GERAN_Receive_UplinkData (geran_Cell24, v_EndOfTBF) {}
      [] G_RLC.receive (car_G_RLC_ControlMsg_IND(geran_Cell24, tsc_PhyCh1, cr_PktCellChangeNotificationAny)){}
    }
    //@siclog "Step 4" siclog@
    G_RLC.send (cas_G_RLC_ControlMsg_REQ_PACCH(geran_Cell24,
                                               tsc_PhyCh1,
                                               1,

                                               tsc_UplinkTFI,
                                               cs_PktCellChangeOrder (cs_GlobalTFIul,
                                                                      cs_EUTRAN_TargetCell (int2bit(v_EUTRA_SysInfo.Eutra[0].Arfcn, 16),
                                                                                            '0'B,
                                                                                            omit,
                                                                                            int2bit (v_EUTRA_SysInfo.Eutra[0].HO.PhysCellId, 9)))));
    //@siclog "Step 5" siclog@
    // Tell the EUTRA component to wait for an RRC Conn Req
    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);
    // Keep going until UE is on cell 1
    alt {
      [not v_EndOfTBF] a_GERAN_Receive_UplinkData (geran_Cell24, v_EndOfTBF) {}
      [] EUTRA.receive(cmr_IRAT_Trigger) {} // Trigger received, so can finish loop now
    }
    // Test now finished
    f_GERAN_TestBody_Set ( false );
    f_GERAN_SS_CellRelease ( geran_Cell24 );
  }
...


Change 2 – Correction to ‘f_GERAN_Trigger_UplinkData ()’

	Function name
	f_GERAN_Trigger_UplinkData ()

	Reason for change
	According to the Main Behaviour, at Step 1, an Uplink Dynamic allocation two phase access procedure as mentioned in TS 51.010 secn 40.4.3.9 should be used to trigger uplink data. 

	Summary of change
	Modified function f_GERAN_Trigger_UplinkData to receive CHANNEL REQUEST from the UE for two phase access and wrote a new function fl_ULTBFTwoPhase in accordance to TS 51.010 secn 40.4.3.9

	TTCN module
	CommonGERAN\ GERAN_CommonFunctions.ttcn

	MCC160 Comment
	


Before change

	...

function f_GERAN_Trigger_UplinkData (GERAN_CellId_Type p_CellId,
                                       integer p_Data,
                                       integer p_Sapi := tsc_LLCSapi_11) runs on GERAN_PTC
  {
    var G_L2_DATAMESSAGE_IND v_ChanReq;
    f_UT_GeranUplinkData (UT, int2str(p_Data), int2str(p_Sapi));
    G_L2.receive (car_G_L2_ACCESS_IND ( p_CellId, tsc_PhyCh0, tsc_RACH, ?, ?, cr_ChanReqOnePhase )) -> value v_ChanReq;
    fl_ULTBFOnePhase (p_CellId, v_ChanReq.access);
  }
...


After change

	...

function f_GERAN_Trigger_UplinkData (GERAN_CellId_Type p_CellId,
                                       integer p_Data,
                                       integer p_Sapi := tsc_LLCSapi_11) runs on GERAN_PTC
  {
    var G_L2_DATAMESSAGE_IND v_ChanReq;
    f_UT_GeranUplinkData (UT, int2str(p_Data), int2str(p_Sapi));
    G_L2.receive (car_G_L2_ACCESS_IND ( p_CellId, tsc_PhyCh0, tsc_RACH, ?, ?, cr_ChanReqTwoPhase )) -> value v_ChanReq;
    fl_ULTBFTwoPhase (p_CellId, v_ChanReq.access);
  }
...


New Function:

	...

function fl_ULTBFTwoPhase(GERAN_CellId_Type p_CellId,
                            G_L2_ACCESS_IND p_AccessInd ) runs on GERAN_PTC
  { 
    /* @sic R5s120050 MCC160 implementation sic@ */
    var G_CPHY_CONFIG_CNF v_G_CL1_ComingFN_CNF;
    var G_RLC_DATAMESSAGE_IND v_G_RLC_ControlMsg_IND;
    var RFN v_StartingTime;
    var B5_Type v_Gamma := f_GERAN_GAMMA_Get(p_CellId);
    var B10_Type v_GPRSFreq := f_GERAN_GPRSFreq_Get(p_CellId);
    var integer v_RR_RA;
    var RFN v_RFN;
    var TLLI v_TLLI;
    v_RFN := p_AccessInd.rfn;
    v_RR_RA := bit2int(p_AccessInd.burst.chReq.estCauRandomRef);
    if ((v_RR_RA < 112) or (v_RR_RA > 119)) {// establishment cause should be two phase access
      f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "wrong establishment cause");
    }
    v_G_CL1_ComingFN_CNF := f_GERAN_CPHY_CONFIG_ComingFN(cas_G_CL1_ComingFN_REQ( p_CellId, tsc_PhyCh1, tsc_PACCH_F));
    v_StartingTime := v_G_CL1_ComingFN_CNF.comingFN;
    // Send IA for Single block allocation
    G_L2.send (cas_G_L2_UNITDATA_REQ (p_CellId,
                                      tsc_PhyCh0,
                                      tsc_AGCH, 15,
                                      cs_G_RFN_Omit,
                                      cas_ImmAss (cs_P_ImmediateAssignment ('0'B,
                                                                            v_GPRSFreq,
                                                                            v_RR_RA,
                                                                            v_RFN,
                                                                            cs_IARO_SingleBlk_PUA( v_Gamma,
                                                                                         (v_StartingTime.t1_ & v_StartingTime.t3 & v_StartingTime.t2))))));  // @sic GP-110839 sic@

    //Configure the Uplink TBF for Dynamic Allocation, USF 0 on timeslot 4
    f_GERAN_RLC_CONFIG_Common(cas_ActivateTBF_UplinkDyn(p_CellId, 1), dontWaitForCNF);  // don't wait for CNF to avoid race condition with incoming car_G_RLC_ControlMsg_IND
    // Now wait for PRR or PMR
    alt {
      [] G_RLC.receive (car_G_RLC_ControlMsg_IND (p_CellId, tsc_PhyCh1, cr_PacketResourceRequest2Phase))
         -> value v_G_RLC_ControlMsg_IND
        {
          v_TLLI := f_GERAN_TLLI_Get(p_CellId);
          G_RLC.send (cas_G_RLC_ControlMsg_REQ_PACCH( p_CellId, tsc_PhyCh1, 1, tsc_UplinkTFI, cs_PacketUplinkAssignment_2phase(v_GPRSFreq, v_TLLI)));
        }
      [] G_RLC.receive (car_G_RLC_ControlMsg_IND(geran_Cell24, tsc_PhyCh1, cr_PktMeasReportAny)) 
        {
          repeat;
        } 
    }
  } // end of f_ULTBFTwoPhase
...


New Templates for function f_ULTBFTwoPhase:

	...

template AccessIndType cr_ChanReqTwoPhase := cr_G_ChannelReq('01110???'B);

...


	...

/*Single block Immediate Assignment*/
  template (value) IARO cs_IARO_SingleBlk_PUA(B5_Type p_Gamma,
                                              B16_Type p_StartTime) :=
  { /* @status    APPROVED */
    constructMask   := '11'B,
    compressInterRATHO := omit,                      // present if constructMask = LL
    reservMask1     := omit,                         // present if constructMask = LH
    reservMask2     := omit,                         // present if reservMask1 = 0
    eGPRS_UplinkAssign := omit,                      // not present if reservMask2 = 1 or reservMask1 = 1 or constructMask <> LH
    multiBlocksPDA  := omit,                         // not present if reservMask2 = 0 or reservMask1 = 1 or constructMask <> LH
    lengthOfFreqParam := omit,                       // present if constructMask = HL
    freqParam_bt    := omit,                         // frequency parameters before time
                                                     // not present if lengthOfFreqParam = 0
    compressInterRATHO1_Type := omit,                     // present if constructMask = HL
    mask1           := '0'B,                         // present if constructMask = HH
    mask2           := '0'B,
    packetUplinkAssign := cs_PktUplinkAss_SingleBlk_IARO(p_Gamma, p_StartTime), // not present if mask1 = 1 or mask2 = 1 or constructMask <> HH
    packetDownlinkAssign := omit,                    // not present if mask1 = 1 or mask2 = 0 or constructMask <> HH
    secondPacketAssign := omit,                       // present if mask1 = 1 and constructMask = HH
    r10Mask         := '0'B,                       // @sic R5s110178 Baseline Moving sic@
    implicitRejectCS := omit,    // not present if r10Mask = L @sic R5s110178 Baseline Moving sic@
    implicitRejectPS := omit    // not present if r10Mask = L @sic R5s110178 Baseline Moving sic@
  };

...


	...

/*PUA for single block allocation*/
  template (value) PacketUplinkAssign cs_PktUplinkAss_SingleBlk_IARO(B5_Type p_Gamma,
                                                           B16_Type p_StartTime) :=
  { /* @status    APPROVED */
    allocMask       := '0'B,                         // 0 -> indicating single block allocation
    tFI_Assignment  := omit,                // temporary flow identifier
                                                     // not present if allocMask = 0
    pOLLING         := omit,                         // not present if allocMask = 0
    spareBit        := omit,                         // not present if allocMask = 0
    uSF             := omit,                   // not present if allocMask = 0 or fixedAllocMask = 1
    uSFGranularity  := omit,                         // not present if allocMask = 0 or fixedAllocMask = 1
    p0Mask          := omit,                         // not present if allocMask = 0 or fixedAllocMask = 1
    p0              := omit,                         // not present if allocMask = 0 or fixedAllocMask = 1 or p0Mask = 0
    pR_MODE         := omit,                         // not present if allocMask = 0 or fixedAllocMask = 1 or p0Mask = 0
    ChannelCodingCommand := omit,                   // not present if allocMask = 0
    tLLI_BlockChannelCodeing := omit,                // not present if allocMask = 0
    alphaMask       :=omit,                         // not present if allocMask = 0
    aLPHA           := omit,                      // not present if allocMask = 0 or alphaMask = 0
    gMMA            := omit,                      // not present if allocMask = 0
    taMask          := omit,                         // not present if allocMask = 0
    tIMING_ADVANCE_INDEX := omit,                    // not present if allocMask = 0 or taMask = 0
    startingTimeMask := omit,                        // not present if allocMask =  0
    tBF_STARTING_TIME := omit,                // not present if allocMask = 0 or startingTimeMask = 0
    alpha1Mask      := '1'B,                         // present if allocMask = 0 single block allocation
    aLPHA1          :=  '0101'B,                        // not present if allocMask = 1 or alpha1Mask = 0
    gMMA1           := p_Gamma,                         // not present if allocMask = 1
    prefixBits      := '01'B,                         // not present if allocMask = 1 the value is 01 if present
    tBF_STARTING_TIME1 := p_StartTime,                      // not present if allocMask = 1
    p02Mask         := omit,                         // not present if allocMask = 1
    p02             := omit,                         // not present if allocMask = 1 or p02Mask = L
    spareBit2       := omit,                         // not present if allocMask = 1 or p02Mask = L
    pR_MODE2        := omit,                         // not present if allocMask = 1 or p02Mask = L
    r99Mask         := '0'B,
    extRAMask       := omit,                         // not present if r99Mask = L
    extRA           := omit,                         // not present if r99Mask = L or extRAMask = 0
    r6Mask          := omit,                         // not present if r99Mask = L
    pfiMask         := omit,                         // not present if r99Mask = L or r6Mask = L
    pFI             := omit                          // not present if r99Mask = L, r6Mask = L or pfiMask = 0
  };

...


	...

  /*Packet Resource Request for Two Phase Access*/
  template GRLCCntrlMsgIndType cr_PacketResourceRequest2Phase :=
  { /* @status    APPROVED */
    pktResReq := {
      msgType    := '000101'B,               // message type   M
      mask1      := '1'B,
      accessType := '00'B, // not present if mask1 = 0
      mask2      := '1'B,
      globalTFI  := omit, // only present if mask2 = 0
      tlli       := *, // only present if mask2 = 1
      mask3      := ?,
      mSRadioAccessCap := cr_MSRadioAccessCapListAny ifpresent, // not present if mask3 = 0
      chReqDescr := cr_ChReqDescr2Phase,
      mask4      := ?,
      changeMark := *, // not present if mask4 = 0
      cValue     := ?,
      mask5      := ?,
      signVar    := *, // not present if mask5 = 0
      tnmask0    := ?,
      iLevelTN0  := *, // not present if tnmask0 = 0
      tnmask1    := ?,
      iLevelTN1  := *, // not present if tnmask1 = 0
      tnmask2    := ?,
      iLevelTN2  := *, // not present if tnmask2 = 0
      tnmask3    := ?,
      iLevelTN3  := *, // not present if tnmask3 = 0
      tnmask4    := ?,
      iLevelTN4  := *, // not present if tnmask4 = 0
      tnmask5    := ?,
      iLevelTN5  := *, // not present if tnmask5 = 0
      tnmask6    := ?,
      iLevelTN6  := *, // not present if tnmask6 = 0
      tnmask7    := ?,
      iLevelTN7  := *, // not present if tnmask7 = 0
      r99mask    := ?,
      mask6      := *, // not present if r99mask = 0
      eGPRSBEPLinkQual := cr_EGPRSBEPLinkQualAny ifpresent, // not present if r99mask or if mask6 = 0
      mask7      := *, // not present if r99mask = 0
      eGPRSTNLinkQual:= cr_EGPRSTNLinkQualAny ifpresent, // not present if r99mask or mask7= 0
      mask8      := *, // not present if r99mask = 0
      pFI        := *, // not present if r99mask or mask8= 0
      addMSRACInfo:= *, // not present if r99mask = 0
      retxPRR    := *, // not present if r99mask = 0
      rel5mask   := *, // not present if r99mask = 0
      mask9      := *, // not present if r99mask or rel5mask = 0
      mask10     := *, // not present if r99mask or rel5mask or mask9 = 0
      gRNTIExt   := *, // not present if r99mask,rel5mask, mask9 or mask10 = 0
      iuModeChReqDescr := cr_IuModeChReqDescrAny ifpresent, // not present if r99mask, rel5mask or mask9 = 0
      hfnmask    := *, // not present if r99mask or rel5mask or mask9 = 0
      hfnlsb     := *, // not present if r99mask or rel5mask or mask9 or hfnmask = 0
      rel6mask   := *, // not present if r99mask or rel5mask or mask9 = 0
      mask11     := *, // not present if r99mask or rel5mask or mask9 or rel6mask = 0
      extChReqDescr := cr_ExtChReqDescrAny ifpresent, // not present if r99mask or rel5mask or mask9 or rel6mask or mask11 = 0
      rel7mask   := *, // not present if r99mask or rel5mask or mask9 or rel6mask = 0
      earlyTBF   := *, // not present if r99mask or rel5mask or mask9 or rel6mask or rel7mask = 0
      mask12     := *, // not present if r99mask or rel5mask or mask9 or rel6mask or rel7mask = 0
      eGPRSBEPLinkQual2 := cr_EGPRSBEPLinkQualAny ifpresent, // not present if r99mask or if mask6 or rel7mask or mask12 = 0
      mask13     := *, // not present if r99mask or rel5mask or mask9 or rel6mask or rel7mask = 0
      eGPRSTNLinkQual2 := cr_EGPRSTNLinkQualAny ifpresent, // not present if r99mask or if mask6 or rel7mask or mask13 = 0
      rel10mask   := *, // not present if r99mask or rel5mask or mask9 or rel6mask or rel7mask = 0 @sic R5s120178 sic@
      lowAccessPrioritySig := *, // not present if r99mask or rel5mask or mask9 or rel6mask or rel7mask or rel10Mask = 0  @sic R5s120178 sic@
      sparePadding := *  // Uplink message
    }
  };
...


	...

/*Channel request for 2 phase access should match parameters sent in PDP Context Accept*/
  template ChReqDescr cr_ChReqDescr2Phase :=
  { /* @status    APPROVED */
    pkThrouputClass := '0101'B,
    radioPriority   := '11'B,
    rlcMode         := '0'B,   // 0 - RLC acknowledged mode 1 - RLC unacknowledged mode
    llcPduType         := ?,  // 0 - LLC PDU is SACK or ACK 1 - LLC PDU is not SACK or ACK
    rlcOctetCount   := ?
  };
...


	...

// Packet UL Assignment for 2 Phase access
  template (value) GRLCCntrlMsgReqType cs_PacketUplinkAssignment_2phase(template (value) B10_Type p_ARFCN, 
                                                                        template (value) O4_Type p_TLLI) :=
  { /* This message is sent on the PCCCH or PACCH by the network to the mobile station to assign uplink resources for One phase TBF.
       TS 3GPP 44.060 clause 11.2.29
       Values are taken from 51010 clause 40.2.3.4 */
    /* @status    APPROVED */
    pktULAssignmt := {
      msgType         := '001010'B,                    // Message type
      pAGE_MODE       := int2bit(tsc_NormalPaging,2),  // Normal paging
      persistenceLevelMask := '0'B,
      persistenceLevel1 := omit,                       // not Present if persistenceLevelMask=0
      persistenceLevel2 := omit,                       // not Present if persistenceLevelMask=0
      persistenceLevel3 := omit,                       // not Present if persistenceLevelMask=0
      persistenceLevel4 := omit,                       // not Present if persistenceLevelMask=0
      globalTFIMask := '1'B,
      globalTFI := omit,
      tLLIMask := '0'B,
      tLLI := p_TLLI,
      tQIMask := omit,
      tQI := omit,
      referenceAddress := omit,
      eGPRSmask      := '0'B,
      channelCodingCommand := int2bit(tsc_CS1,2),      // not present of EGPRPSmask=1
      tLLIBlockChannelCoding := '0'B,                  // 0 - the mobile station shall use CS-1 in GPRS TBF mode and MCS-1 in EGPRS TBF mode.
                                                       // 1 - the mobile station shall use the value commanded in the CHANNEL_CODING_COMMAND or EGPRS_CHANNEL_CODING_COMMAND field.
      packetTimingAdvance := cs_PacketTimingAdvance('1'B, '000000'B),
      freqParamMask   := '1'B,
      freqParam       := cs_FreqParam_PDTCH(p_ARFCN),   // not present if freqParamMask=0
      allocationMask  := '01'B,
      dynamicAllocation := cs_DynamicAllocationPacket, // present if allocationMask=01
      singleBlockAllocation := omit,                   // present if allocationMask=10
      r99Mask         := '1'B,
      packetExtTAmask := '0'B,                         // not present if r99Mask=0
      packetExtendedTA := omit,                        // not present if r99Mask=0 and packetExtTImask=0
      r5Mask          := omit,
      gRNTIMask       := omit,                         // not present if r99Mask=0 and r5Mask=0
      gRNTIExtension  := omit,                         // not present if r99Mask=0, r5Mask=0 and gRNTIMask=0
      rBIdMask        := omit,                         // not present if r99Mask=0 and r5Mask=0
      rBId            := omit,                         // not present if r99Mask=0, r5Mask=0 and rBIdMask=0
      r6Mask          := omit,                         // not present if r99Mask=0, r5Mask=0
      pfiMask         := omit,                         // not present if r99Mask=0, r5Mask=0 and r6Mask = 0
      pFI             := omit,                         // not present if r99Mask=0, r5Mask=0, r6Mask = 0 and pfiMask = 0
      rlcModeMask     := omit,                         // not present if r99Mask=0, r5Mask=0 and r6Mask = 0
      rlcMode         := omit,                         // not present if r99Mask=0, r5Mask=0, r6Mask = 0 and rlcModeMask = 0
      mask            := omit,
      eGPRSmessage    := omit,                         // not present of EGPRPSmask=0
      r10Mask         := omit,
      emsrAddPFC1     := omit,
      emsrAddPFC2     := omit,
      emsrAddPFC3     := omit,
      dualCarrier     := omit
    }
  };
...


Change 3 – Correction to ‘cr_PktCellChangeNotificationAny’

	Template name
	cr_PktCellChangeNotificationAny

	Reason for change
	According to TS 44.060 secn 11.2.3a, IE 3G CCN Measurement Report is present only if the mask bit for R6 is set to 1. Hence, it should be set as optional. 

	Summary of change
	Modified the template cr_PktCellChangeNotificationAny to set the IE ccn3GMeasReport as ‘*‘ instead of ‘?’.

	TTCN module
	CommonGERAN\ GERAN_Templates.ttcn

	MCC160 Comment
	


Before change

	...

  template GRLCCntrlMsgIndType cr_PktCellChangeNotificationAny :=
  {
    pktCellChangeNotification := {
      msgType := '001100'B,               // Message type
      gLOBAL_TFI   := ?,
      gsmMask := ?,
      arfcn  := *, // not present if gsmMask = 1
      bsic  := *,   // not present if gsmMask = 1
      rel6Mask  := *, // not present if gsmMask = 0
      target3GCell  := *, // not present if gsmMask = 0 or rel6Mask = 1
      rel8Mask  := *, // not present if gsmMask = 0 or rel6Mask = 0
      arfcnMask   := *,// not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 1
      earfcn  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 1 or arfcnMask = 0
      ebsic  := *,   // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 1 or arfcnMask = 0
      e3GMask  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 1
      etarget3GCell  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 1 or e3GMask = 0
      eTargetMask  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 1
      EUTRAN_TargetCell  := *,// not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 1 or eTargetMask = 0
      eutranRepQuantity  := *,// not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 1 or eTargetMask = 0
      eCCNRptMask  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 1
      eCCNMeasReport  := *,// not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 1 or eCCNRptMask = 0
      rel9Mask  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 0
      csgMask  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 0 or rel9Mask = 1
      uCSGMeasRpt  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 0 or rel9Mask = 1 or  or csgMask = 0
      eCSGMeasRpt  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 0 or rel9Mask = 1 or csgMask = 1
      eCCNMask  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 0 or rel9Mask = 1
      eCCNMeasRpt  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 0 or rel9Mask = 1 or eCCNMask = 0
      baMask := ?,
      baUsed  := *,    // not present if baMask = 1
      psi3ChangeMark  := *,    // not present if baMask = 0
      pmoUsed := ?,
      pccnSending := ?,
      nCMeasReport := ?,
      r6Mask := ?,
      ba3GMask  := *, // not present if r6Mask = 0
      ba3GUsed   := *,    // not present if r6Mask = 0 or ba3GMask = 0
      n3G  := *, // not present if r6Mask = 0
      ccn3GMeasReport := ?,
      r10Mask := *, // not present if r6Mask = 0 @sic R5s120178 Baseline Moving sic@
      csgDiscriminator := *, // not present if r6Mask = 0 or r10Mask = 0 @sic R5s120178 Baseline Moving sic@
      paddingBits  := *                   // uplink, so have to receive this in TTCN
    }
  };
...


After change

	...

 template GRLCCntrlMsgIndType cr_PktCellChangeNotificationAny :=
  {
    pktCellChangeNotification := {
      msgType := '001100'B,               // Message type
      gLOBAL_TFI   := ?,
      gsmMask := ?,
      arfcn  := *, // not present if gsmMask = 1
      bsic  := *,   // not present if gsmMask = 1
      rel6Mask  := *, // not present if gsmMask = 0
      target3GCell  := *, // not present if gsmMask = 0 or rel6Mask = 1
      rel8Mask  := *, // not present if gsmMask = 0 or rel6Mask = 0
      arfcnMask   := *,// not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 1
      earfcn  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 1 or arfcnMask = 0
      ebsic  := *,   // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 1 or arfcnMask = 0
      e3GMask  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 1
      etarget3GCell  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 1 or e3GMask = 0
      eTargetMask  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 1
      EUTRAN_TargetCell  := *,// not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 1 or eTargetMask = 0
      eutranRepQuantity  := *,// not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 1 or eTargetMask = 0
      eCCNRptMask  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 1
      eCCNMeasReport  := *,// not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 1 or eCCNRptMask = 0
      rel9Mask  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 0
      csgMask  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 0 or rel9Mask = 1
      uCSGMeasRpt  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 0 or rel9Mask = 1 or  or csgMask = 0
      eCSGMeasRpt  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 0 or rel9Mask = 1 or csgMask = 1
      eCCNMask  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 0 or rel9Mask = 1
      eCCNMeasRpt  := *, // not present if gsmMask = 0 or rel6Mask = 0 or rel8Mask = 0 or rel9Mask = 1 or eCCNMask = 0
      baMask := ?,
      baUsed  := *,    // not present if baMask = 1
      psi3ChangeMark  := *,    // not present if baMask = 0
      pmoUsed := ?,
      pccnSending := ?,
      nCMeasReport := ?,
      r6Mask := ?,
      ba3GMask  := *, // not present if r6Mask = 0
      ba3GUsed   := *,    // not present if r6Mask = 0 or ba3GMask = 0
      n3G  := *, // not present if r6Mask = 0
      ccn3GMeasReport := *, // not present if r6Mask = 0
      r10Mask := *, // not present if r6Mask = 0 @sic R5s120178 Baseline Moving sic@
      csgDiscriminator := *, // not present if r6Mask = 0 or r10Mask = 0 @sic R5s120178 Baseline Moving sic@
      paddingBits  := *                   // uplink, so have to receive this in TTCN
    }
  }; 

...


Change 4 – Correction to ‘f_TC_6_2_3_24_EUTRA ()’
	Function name
	f_TC_6_2_3_24_EUTRA ()

	Reason for change
	1. The variables v_SecurityParams and v_GutiParams are being assigned with values even before the function for initialization for E-UTRA Cell 1 is called.
2. The variable v_EPSUpdate_TypeValue is being assigned a value of  tsc_EPSUpdate_Combined_IMSIAttach without the necessary check for combined TRACKING AREA/LOCATION AREA updating.

	Summary of change
	1. Assigned variables v_SecurityParams and v_GutiParams with correct values at the appropriate places.
2. Added an appropriate check to assign the variable v_EpsUpdate_TypeValue.

	TTCN module
	6_2\InterRat_CellReSelection_GtoE.ttcn

	MCC160 Comment
	


Before change

	...

function f_TC_6_2_3_24_EUTRA ( ) runs on EUTRA_PTC {
    var boolean v_CSInOtherRAT := false;
    var NAS_MSG_Indication_Type v_NasInd;
    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();
    var NasCount_Type v_NasCountUL;
    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti ( eutra_Cell1 );
    var B3_Type v_EpsUpdate_TypeValue := f_GetEPSTAUType(NORMAL); // @sic R5s100029, R5s110176 sic@
    var B3_Type v_EpsUpdate_Result := v_EpsUpdate_TypeValue; // To be used in Accept message
    var template (present) EPS_BearerContextStatus v_EPS_ContextStatus_Rx := cr_508_EPS_BearerContextStatusTAUReq; // @sic R5s100608, R5-110746 sic@
    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus_Tx;
    var template AdditionalUpdateType v_AdditionalUpdateType_Expected := f_GetAdditionalUpdateType (NORMAL); // @sic R5s110176 sic@
    var template (present) NAS_MSG_Indication_Type v_TAU_Expected, v_TAU_NoProtection, v_TAU_StartInOtherRAT;
    //Initialise all cells, security and mobile parameters
    f_EUTRA_Init( c5 );//EUTRA-GERAN configuration
    f_EUTRA_CellInfo_InitMaxReferencePower (eutra_Cell1, -60);
    //Create and configure Cell 1
    f_EUTRA_CellConfig_Def ( eutra_Cell1 );
    //Send cell information to Geran side to update SI on preamble
    f_IRAT_Send_EUTRA_SysInfoToOtherPort(GERAN, f_EUTRA_CellInfo_GetEARFCN_DL(eutra_Cell1),
                                         f_ConvertDL_BandwidthToInteger (f_EUTRA_CellInfo_GetDL_ChBandwidth (eutra_Cell1)),
                                         cs_EUTRA_HOInfo (f_ConvertUL_BandwidthToInteger (f_EUTRA_CellInfo_GetUL_ChBandwidth (eutra_Cell1)),
                                                          f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1),
                                                          f_EUTRA_CellInfo_GetRootSequenceIndex (eutra_Cell1)));
    //Receive Authentication parameters updated in the RAU on GERAN side
    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(GERAN);
    f_EUTRA_TestBody_Set ( true );
    //@siclog "Step 2" siclog@
    f_IRAT_WaitForCoOrd_Trigger ( GERAN );
    //Turn cell on
    f_EUTRA_SetCellPower (eutra_Cell1, -60);
    if (v_EpsUpdate_TypeValue != tsc_EpsUpdate_TaUpdate) { // TA Only;  @sic R5-110746 sic@
        v_EpsUpdate_TypeValue := tsc_EpsUpdate_Combined_IMSIAttach;

    }
    v_TAU_NoProtection := cr_NAS_Indication(tsc_SHT_NoSecurityProtection,
                                           cr_TAU_RequestIdleMobilityDiffRAT ( v_EpsUpdate_TypeValue,
                                                                               '111'B,
                                                                               v_EPS_ContextStatus_Rx,
                                                                               v_AdditionalUpdateType_Expected,
                                                                               ? ));
    v_TAU_StartInOtherRAT := cr_NAS_Indication(tsc_SHT_IntegrityProtected,
                                              cr_TAU_RequestIdleMobilityDiffRAT ( v_EpsUpdate_TypeValue,
                                                                                  ?,
                                                                                  v_EPS_ContextStatus_Rx,
                                                                                  v_AdditionalUpdateType_Expected,
                                                                                  ? ));
    if (v_SecurityParams.KSIsgsn == '111'B ) {// No current EPS security context stored
      //@sic R5-110746 Don't care if UE is sending mapped or native sic@
      v_TAU_Expected := (v_TAU_NoProtection, v_TAU_StartInOtherRAT);
    }
    //@siclog "Step 5" siclog@
    f_IRAT_WaitForCoOrd_Trigger ( GERAN );
    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1, mo_Signalling);    // @sic R5-120721 change 3 sic@
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5");
    //@siclog "Step 6" siclog@
    f_EUTRA_RRC_ConnectionSetup_Def (eutra_Cell1);
    //@siclog "Step 7" siclog@
    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_Cell1,                                       // @sic R5s120121 MCC160 implementation sic@ 
                                                       cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),
                                                       v_TAU_Expected);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 7");
    f_EUTRA_TestBody_Set ( false );
    v_SecurityParams.NonceUE := oct2bit(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.nonce.nonceValue);
    v_SecurityParams := f_EUTRA_MappedContext_InitNAS(v_SecurityParams, v_SecurityParams.AuthParams);
    v_SecurityParams.NAS_SecurityCap := f_EUTRA_FillNAS_SecurityCapability_TAU(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST); // @sic R5s110210 sic@
    v_NasCountUL := f_EUTRA_NAS_ActivateSecurity_MappedContext(eutra_Cell1, v_SecurityParams);
    v_EPS_ContextStatus_Tx := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept
    v_SecurityParams.NAS_SecurityCap := f_EUTRA_FillNAS_SecurityCapability_TAU(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST);
    f_EUTRA_Security_Set(v_SecurityParams);
    f_EUTRA_TrackingAreaUpdate_Accept_Complete ( eutra_Cell1,
                                                   v_GutiParams,
                                                   v_EpsUpdate_Result,
                                                   v_EPS_ContextStatus_Tx,
                                                   v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType,
                                                   NORMAL);
    // Tell the GERAN component that the test is now over
    f_IRAT_SendCoOrd (GERAN, cms_IRAT_Trigger);
    // Switch off UE and expect Detach Request
    f_EUTRA_Postamble ( eutra_Cell1, E2_CONNECTED );
  }
...


After change

	...

function f_TC_6_2_3_24_EUTRA ( ) runs on EUTRA_PTC {
    var boolean v_CSInOtherRAT := false;
    var NAS_MSG_Indication_Type v_NasInd;
    var EUTRA_SecurityParams_Type v_SecurityParams;
    var NasCount_Type v_NasCountUL;
    var GutiParameters_Type v_GutiParams;
    var B3_Type v_EpsUpdate_TypeValue := f_GetEPSTAUType(NORMAL); // @sic R5s100029, R5s110176 sic@
    var B3_Type v_EpsUpdate_Result := v_EpsUpdate_TypeValue; // To be used in Accept message
    var template (present) EPS_BearerContextStatus v_EPS_ContextStatus_Rx := cr_508_EPS_BearerContextStatusTAUReq; // @sic R5s100608, R5-110746 sic@
    var template (omit) EPS_BearerContextStatus v_EPS_ContextStatus_Tx;
    var template AdditionalUpdateType v_AdditionalUpdateType_Expected := f_GetAdditionalUpdateType (NORMAL); // @sic R5s110176 sic@
    var template (present) NAS_MSG_Indication_Type v_TAU_Expected, v_TAU_NoProtection, v_TAU_StartInOtherRAT;
    //Initialise all cells, security and mobile parameters
    f_EUTRA_Init( c5 );//EUTRA-GERAN configuration
    f_EUTRA_CellInfo_InitMaxReferencePower (eutra_Cell1, -60);
    //Create and configure Cell 1
    f_EUTRA_CellConfig_Def ( eutra_Cell1 );
    //Send cell information to Geran side to update SI on preamble
    f_IRAT_Send_EUTRA_SysInfoToOtherPort(GERAN, f_EUTRA_CellInfo_GetEARFCN_DL(eutra_Cell1),
                                         f_ConvertDL_BandwidthToInteger (f_EUTRA_CellInfo_GetDL_ChBandwidth (eutra_Cell1)),
                                         cs_EUTRA_HOInfo (f_ConvertUL_BandwidthToInteger (f_EUTRA_CellInfo_GetUL_ChBandwidth (eutra_Cell1)),
                                                          f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1),
                                                          f_EUTRA_CellInfo_GetRootSequenceIndex (eutra_Cell1)));
    //Receive Authentication parameters updated in the RAU on GERAN side
    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(GERAN);
    f_EUTRA_TestBody_Set ( true );
    //@siclog "Step 2" siclog@
    f_IRAT_WaitForCoOrd_Trigger ( GERAN );
    //Turn cell on
    f_EUTRA_SetCellPower (eutra_Cell1, -60);
    if (v_EpsUpdate_TypeValue != tsc_EpsUpdate_TaUpdate) { // TA Only;  @sic R5-110746 sic@
       if (v_CSInOtherRAT) {
        v_EpsUpdate_TypeValue := tsc_EpsUpdate_Combined_IMSIAttach;
      } else {
        v_EpsUpdate_TypeValue := tsc_EpsUpdate_Combined_TaLaUpdate;
      }
    }
    v_TAU_NoProtection := cr_NAS_Indication(tsc_SHT_NoSecurityProtection,
                                           cr_TAU_RequestIdleMobilityDiffRAT ( v_EpsUpdate_TypeValue,
                                                                               '111'B,
                                                                               v_EPS_ContextStatus_Rx,
                                                                               v_AdditionalUpdateType_Expected,
                                                                               ? ));
    v_TAU_StartInOtherRAT := cr_NAS_Indication(tsc_SHT_IntegrityProtected,
                                              cr_TAU_RequestIdleMobilityDiffRAT ( v_EpsUpdate_TypeValue,
                                                                                  ?,
                                                                                  v_EPS_ContextStatus_Rx,
                                                                                  v_AdditionalUpdateType_Expected,
                                                                                  ? ));
    //Get the security params for E-UTRA Cell 1 here
    v_SecurityParams := f_EUTRA_Security_Get();
    if (v_SecurityParams.KSIsgsn == '111'B ) {// No current EPS security context stored
      //@sic R5-110746 Don't care if UE is sending mapped or native sic@
      v_TAU_Expected := (v_TAU_NoProtection, v_TAU_StartInOtherRAT);
    }
    //@siclog "Step 5" siclog@
    f_IRAT_WaitForCoOrd_Trigger ( GERAN );
    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1, mo_Signalling);    // @sic R5-120721 change 3 sic@
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5");
    //@siclog "Step 6" siclog@
    f_EUTRA_RRC_ConnectionSetup_Def (eutra_Cell1);
    //@siclog "Step 7" siclog@
    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_Cell1,                                       // @sic R5s120121 MCC160 implementation sic@ 
                                                       cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),
                                                       v_TAU_Expected);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 7");
    f_EUTRA_TestBody_Set ( false );
    v_SecurityParams.NonceUE := oct2bit(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.nonce.nonceValue);
    v_SecurityParams := f_EUTRA_MappedContext_InitNAS(v_SecurityParams, v_SecurityParams.AuthParams);
    v_SecurityParams.NAS_SecurityCap := f_EUTRA_FillNAS_SecurityCapability_TAU(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST); // @sic R5s110210 sic@
    v_NasCountUL := f_EUTRA_NAS_ActivateSecurity_MappedContext(eutra_Cell1, v_SecurityParams);
    v_EPS_ContextStatus_Tx := v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus; // Send this back in the Accept
    v_SecurityParams.NAS_SecurityCap := f_EUTRA_FillNAS_SecurityCapability_TAU(v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST);
    f_EUTRA_Security_Set(v_SecurityParams);
    //Get the GUTI params here
    v_GutiParams:= f_EUTRA_CellInfo_GetGuti ( eutra_Cell1 ); 
    f_EUTRA_TrackingAreaUpdate_Accept_Complete ( eutra_Cell1,
                                                   v_GutiParams,
                                                   v_EpsUpdate_Result,
                                                   v_EPS_ContextStatus_Tx,
                                                   v_NasInd.Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType,
                                                   NORMAL);
    // Tell the GERAN component that the test is now over
    f_IRAT_SendCoOrd (GERAN, cms_IRAT_Trigger);
    // Switch off UE and expect Detach Request
    f_EUTRA_Postamble ( eutra_Cell1, E2_CONNECTED );
  }
...


5. Execution Log Files

5.1 Nvidia ICERA 410 UE 

The Nvidia ICERA 410 UE passed this test case on R&S CMW500 with NAS and AS Snow 3G integrity and ciphering algorithm in LTE FDD band 1 and GSM_P900 band. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files: 
tc_6_2_3_24_Nvidia_FDD_band_4_GSM_850.log   
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)

In the log file (in .txt format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

References

	[1]
	R5s120522: Supporting information for agreement of TC 6.2.3.24. This archive comprises:

- text format execution log files

- draft_GP-12nnnn_G#55_RS_6-2-3-24.doc

